
Intercept and transform Ecto repository operations using a middleware pipeline pattern.

EctoMiddleware provides a clean, composable way to add cross-cutting concerns to your Ecto operations. Inspired by Absinthe's middleware and Plug, it allows you to transform data before and after database operations, or replace operations entirely.
Features
	Transform data before it reaches the database (normalization, validation, enrichment)
	Transform data after database operations (logging, notifications, caching)
	Replace operations entirely (soft deletes, read-through caching, authorization)
	Halt execution with authorization checks or validation failures
	Telemetry integration for observability
	Backwards compatible with v1.x middleware

Installation
Add ecto_middleware to your dependencies in mix.exs:
def deps do
  [
    {:ecto_middleware, "~> 2.0"}
  ]
end
Quick Start
1. Enable middleware in your Repo
defmodule MyApp.Repo do
  use Ecto.Repo,
    otp_app: :my_app,
    adapter: Ecto.Adapters.Postgres

  use EctoMiddleware.Repo

  # Define middleware for specific operations
  @impl EctoMiddleware.Repo
  def middleware(action, resource) when is_insert(action, resource) do
    [NormalizeEmail, HashPassword, AuditLog]
  end

  def middleware(action, resource) when is_delete(action, resource) do
    [SoftDelete, AuditLog]
  end

  def middleware(_action, _resource) do
    [AuditLog]
  end
end
2. Write middleware
defmodule NormalizeEmail do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_before(changeset, _resolution) do
    case Ecto.Changeset.fetch_change(changeset, :email) do
      {:ok, email} ->
        {:cont, Ecto.Changeset.put_change(changeset, :email, String.downcase(email))}

      :error ->
        {:cont, changeset}
    end
  end
end
3. Use your Repo as normal
# Middleware runs automatically!

%User{}
|> User.changeset(%{name: "Alice", email: "ALICE@EXAMPLE.COM"})
|> MyApp.Repo.insert()
# => Email is normalized to "alice@example.com" before insertion
Core Concepts
The API
Implement process_before/2 to transform data before the database operation:
defmodule AddTimestamp do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_before(changeset, _resolution) do
    {:cont, Ecto.Changeset.put_change(changeset, :processed_at, DateTime.utc_now())}
  end
end
Implement process_after/2 to process data after the database operation:
defmodule NotifyAdmin do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_after({:ok, user} = result, _resolution) do
    Task.start(fn -> send_notification(user) end)
    {:cont, result}
  end

  def process_after(result, _resolution) do
    {:cont, result}
  end
end
Implement process/2 to wrap around the entire operation:
defmodule MeasureLatency do
  use EctoMiddleware

  @impl EctoMiddleware
  def process(resource, resolution) do
    start_time = System.monotonic_time()

    # Yields control to the next middleware (or Repo operation) in the chain
    # before resuming here.
    {result, _updated_resolution} = yield(resource, resolution)

    duration = System.monotonic_time() - start_time
    Logger.info("#{resolution.action} took #{duration}ns")

    result
  end
end
Important: When using process/2, you must call yield/2 to continue the middleware chain.
Halting Execution
Return {:halt, value} to stop the middleware chain and return a value immediately:
defmodule RequireAuth do
  use EctoMiddleware

  @impl EctoMiddleware
  def process(resource, resolution) do
    if authorized?(resolution) do
      {result, _} = yield(resource, resolution)
      result
    else
      {:halt, {:error, :unauthorized}}
    end
  end

  defp authorized?(resolution) do
    get_private(resolution, :current_user) != nil
  end
end
Guards for Operation Detection
EctoMiddleware provides guards to detect operations, especially useful for insert_or_update:
defmodule ConditionalMiddleware do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_before(changeset, resolution) when is_insert(changeset, resolution) do
    {:cont, add_created_metadata(changeset)}
  end

  def process_before(changeset, resolution) when is_update(changeset, resolution) do
    {:cont, add_updated_metadata(changeset)}
  end

  def process_before(changeset, _resolution) do
    {:cont, changeset}
  end
end
Return Value Conventions
process_before/2 and process_after/2
Returning bare values from middleware is supported, but to be explicit, return one of:
	{:cont, value} - Continue to next middleware
	{:halt, value} - Stop execution, return value
	{:cont, value, updated_resolution} - Continue with updated resolution
	{:halt, value, updated_resolution} - Stop with updated resolution

Bare values are always treated as {:cont, value}.
Telemetry
EctoMiddleware emits telemetry events for observability:
Pipeline Events
	[:ecto_middleware, :pipeline, :start] - Pipeline execution starts
	Measurements: %{system_time: integer()}
	Metadata: %{repo: module(), action: atom(), pipeline_id: reference()}


	[:ecto_middleware, :pipeline, :stop] - Pipeline execution completes
	Measurements: %{duration: integer()}
	Metadata: %{repo: module(), action: atom()}


	[:ecto_middleware, :pipeline, :exception] - Pipeline execution fails
	Measurements: %{duration: integer()}
	Metadata: %{repo: module(), action: atom(), kind: atom(), reason: term()}



Middleware Events
	[:ecto_middleware, :middleware, :start] - Individual middleware starts
	Measurements: %{system_time: integer()}
	Metadata: %{middleware: module(), pipeline_id: reference()}


	[:ecto_middleware, :middleware, :stop] - Individual middleware completes
	Measurements: %{duration: integer()}
	Metadata: %{middleware: module(), result: :cont | :halt}



	[:ecto_middleware, :middleware, :exception] - Individual middleware fails
	Measurements: %{duration: integer()}
	Metadata: %{middleware: module(), kind: atom(), reason: term()}



Example handler:
:telemetry.attach(
  "log-slow-middleware",
  [:ecto_middleware, :middleware, :stop],
  fn _event, %{duration: duration}, %{middleware: middleware}, _config ->
    if duration > 1_000_000 do  # 1ms
      Logger.warn("Slow middleware: #{inspect(middleware)} took #{duration}ns")
    end
  end,
  nil
)
Migration from V1
V1 middleware continue to work but emit deprecation warnings. See the Migration Guide for details.
Key differences:
	Use EctoMiddleware.Repo instead of EctoMiddleware in Repo modules
	Implement process_before/2, process_after/2, or process/2 instead of middleware/2
	No need for EctoMiddleware.Super marker
	Return {:cont, value} or {:halt, value} instead of bare values

Silencing Deprecation Warnings
During migration, you can silence warnings temporarily:
# In config/config.exs
config :ecto_middleware, :silence_deprecation_warnings, true
This should only be used during migration - deprecated APIs will be removed in v3.0.
Example Usage in the Wild
EctoHooks - Adds before_* and after_* callbacks to Ecto schemas, similar to the old Ecto.Model callbacks. Implemented entirely using EctoMiddleware.
See the implementation for real-world middleware examples.
Documentation
	API Documentation
	Migration Guide (v1 → v2)

Contributing
Contributions are welcome! Please feel free to submit a Pull Request.
License
MIT License. See LICENSE for details.


  

    Migrating from V1 to V2

Overview
V2 introduces a cleaner middleware API that removes the need for EctoMiddleware.Super and gives middleware full control over execution flow. V1 middleware continue to work but emit deprecation warnings.
This guide will help you migrate your V1 middleware to V2. We'll use EctoHooks as a real-world example of V1 middleware.
What V1 Middleware Looks Like
Here's a V1 middleware from EctoHooks that runs before database operations:
defmodule EctoHooks do
  defmacro __using__(_opts) do
    quote do
      # ... other code ...
      def middleware(_action, _resource) do
        [
          EctoHooks.Middleware.Before,
          EctoMiddleware.Super,
          EctoHooks.Middleware.After
        ]
      end
    end
  end
end

defmodule EctoHooks.Middleware.Before do
  @behaviour EctoMiddleware

  @impl EctoMiddleware
  def middleware(resource, resolution) when resolution.action in [:insert, :insert!] do
    EctoHooks.before_insert(resource, resolution.action)
  end

  def middleware(resource, resolution) when resolution.action in [:update, :update!] do
    EctoHooks.before_update(resource, resolution.action)
  end

  def middleware(resource, _resolution) do
    resource
  end
end
V1 middleware had limitations:
	Required EctoMiddleware.Super to determine if a middleware is run before/after repo operation
	Couldn't halt execution cleanly
	No control over middleware flow
	Couldn't easily wrap operations (before + after in same middleware)
	Couldn't override repo action entirely

Breaking Changes
1. Repo Setup
Before (V1):
defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app
  use EctoMiddleware  # Deprecated
end
After (V2):
defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app
  use EctoMiddleware.Repo
end
2. yield/2 Return Value
If you're using a middleware intended to run before the repo operation, you need to make the following change:
Before (V1):
def middleware(resource, resolution) do
  result = yield(resource, resolution)
  transform(result)
end
After (V2):
@impl EctoMiddleware
def process_before(resource, resolution) do
  {result, _updated_resolution} = yield(resource, resolution)
  {:cont, transform(result)}
end
Ditto for after middleware:
Before (V1):
def middleware(result, resolution) do
  result = yield(result, resolution)
  transform(result)
end
After (V2):
@impl EctoMiddleware
def process_after(result, resolution) do
  {result, _updated_resolution} = yield(result, resolution)
  {:cont, transform(result)}
end
If your middleware is intended to ship with both before and after logic, you can now do that in a single middleware using process/2:
@impl EctoMiddleware
def process(resource, resolution) do
  # Before logic
  transformed = transform_before(resource)

  # Yield to repo operation
  {result, _updated_resolution} = yield(transformed, resolution)

  # After logic
  final_result = transform_after(result)
end
Migration Strategies
For most V1 middleware, you can use a simple pattern. EctoMiddleware includes V1 wrapper implementations you can reference:
	EctoMiddleware.V1.Before - Wraps V1 before middleware
	EctoMiddleware.V1.After - Wraps V1 after middleware

These show a verbose but generalizable pattern that works for any V1 middleware.
Real-World Examples
Example 1: Simple Transformation
V1:
defmodule MyApp.Middleware.AddTimestamp do
  @behaviour EctoMiddleware

  def middleware(changeset, _resolution) do
    Ecto.Changeset.put_change(changeset, :updated_at, DateTime.utc_now())
  end
end

# In Repo:
def middleware(:update, _resource) do
  [AddTimestamp, EctoMiddleware.Super]
end
V2:
defmodule MyApp.Middleware.AddTimestamp do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_before(changeset, _resolution) do
    {:cont, Ecto.Changeset.put_change(changeset, :updated_at, DateTime.utc_now())}
  end
end

# In Repo:
@impl EctoMiddleware.Repo
def middleware(:update, _resource) do
  [AddTimestamp]  # No Super needed
end
Example 2: Result Processing
V1:
defmodule MyApp.Middleware.EnrichUser do
  @behaviour EctoMiddleware

  def middleware(user, _resolution) do
    %{user | full_name: "#{user.first_name} #{user.last_name}"}
  end
end

# In Repo:
def middleware(:get, User) do
  [EctoMiddleware.Super, EnrichUser]
end
V2:
defmodule MyApp.Middleware.EnrichUser do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_after({:ok, user}, _resolution) do
    {:cont, {:ok, %{user | full_name: "#{user.first_name} #{user.last_name}"}}}
  end

  def process_after(other, _resolution), do: {:cont, other}
end

# In Repo:
def middleware(:get, User) do
  [EnrichUser]  # No Super needed
end
Example 3: Before + After (V2 Improvement)
V1 (Required Two Middleware):
defmodule MyApp.Middleware.LogBefore do
  @behaviour EctoMiddleware
  def middleware(resource, resolution) do
    Logger.info("Before #{resolution.action}")
    resource
  end
end

defmodule MyApp.Middleware.LogAfter do
  @behaviour EctoMiddleware
  def middleware(result, resolution) do
    Logger.info("After #{resolution.action}: #{inspect(result)}")
    result
  end
end

# In Repo:
def middleware(:insert, _) do
  [LogBefore, EctoMiddleware.Super, LogAfter]
end
V2 (Single Middleware):
defmodule MyApp.Middleware.Logger do
  use EctoMiddleware

  @impl EctoMiddleware
  def process(resource, resolution) do
    Logger.info("Before #{resolution.action}")
    {result, _updated_resolution} = yield(resource, resolution)
    Logger.info("After #{resolution.action}: #{inspect(result)}")
    result
  end
end

# In Repo:
def middleware(:insert, _) do
  [Logger]  # Single middleware!
end
Return Value Conventions
process_before/2 and process_after/2
Must return {:cont, value} to continue or {:halt, value} to stop:
def process_before(changeset, _resolution) do
  if valid?(changeset) do
    {:cont, transform(changeset)}
  else
    {:halt, {:error, :invalid}}
  end
end
For convenience, you can return bare values, but this will emit deprecation warnings:
def process_before(changeset, _resolution) do
  transform(changeset)  # Works but warns
end
process/2
Can return anything. The value you return is what gets passed up the chain:
def process(resource, resolution) do
  {result, _} = yield(resource, resolution)
  transform(result)  # This is what the next middleware sees
end
Use {:halt, value} to explicitly signal halting (optional but recommended for clarity):
def process(resource, resolution) do
  if should_halt?(resource) do
    {:halt, {:error, :stopped}}  # Explicit
  else
    {result, _} = yield(resource, resolution)
    result
  end
end
Reference Implementations
For a complete, generalizable pattern that works for any V1 middleware, see the V1 wrapper implementations in EctoMiddleware:
	EctoMiddleware.V1.Before - Shows how V1 before middleware is wrapped in V2
	EctoMiddleware.V1.After - Shows how V1 after middleware is wrapped in V2

These implementations are verbose but demonstrate the pattern you can follow to migrate any V1 middleware. However, many middleware can be simplified by using V2's process/2 callback instead of separate before/after middleware.
Removed Features
EctoMiddleware.Super
Not needed in V2. Just list your middleware without Super:
# V1
def middleware(:insert, _resource) do
  [Before1, Before2, EctoMiddleware.Super, After1, After2]
end

# V2
def middleware(:insert, _resource) do
  [Before1, Before2, After1, After2]
end
Resolution Fields (Deprecated)
The following resolution fields will be removed in v3.0:
	before_input
	before_output
	after_input
	after_output

Don't rely on these in new code. Use resolution.private to pass data between middleware instead.
Silencing Deprecation Warnings
During migration, you may want to silence warnings temporarily:
# In config/config.exs
config :ecto_middleware, :silence_deprecation_warnings, true
This should only be used during migration - deprecated APIs will be removed in v3.0.
Testing Your Migration
	Update your Repo to use EctoMiddleware.Repo
	Run your test suite - V1 middleware should still work with warnings
	Migrate middleware one at a time, running tests after each
	Once all middleware are migrated, warnings should stop

Need Help?
See the API documentation for detailed examples and common patterns.

Behaviour for creating middleware that intercepts and transforms Ecto repository operations.
EctoMiddleware provides a composable way to add cross-cutting concerns to your Ecto
operations using a middleware pipeline pattern, similar to Plug or Absinthe middleware.
Quick Example
defmodule NormalizeEmail do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_before(changeset, _resolution) do
    case Ecto.Changeset.fetch_change(changeset, :email) do
      {:ok, email} ->
        {:cont, Ecto.Changeset.put_change(changeset, :email, String.downcase(email))}
      :error ->
        {:cont, changeset}
    end
  end
end

defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app
  use EctoMiddleware.Repo

  @impl EctoMiddleware.Repo
  def middleware(action, resource) when is_insert(action, resource) do
    [NormalizeEmail, HashPassword, AuditLog]
  end

  def middleware(_action, _resource), do: [AuditLog]
end
Core Concepts
The API
Implement process_before/2 to transform data before the database operation:
defmodule AddTimestamp do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_before(changeset, _resolution) do
    {:cont, Ecto.Changeset.put_change(changeset, :processed_at, DateTime.utc_now())}
  end
end
Implement process_after/2 to process data after the database operation:
defmodule NotifyAdmin do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_after({:ok, user} = result, _resolution) do
    Task.start(fn -> send_notification(user) end)
    {:cont, result}
  end

  def process_after(result, _resolution), do: {:cont, result}
end
Implement process/2 to wrap around the entire operation:
defmodule MeasureLatency do
  use EctoMiddleware

  @impl EctoMiddleware
  def process(resource, resolution) do
    start_time = System.monotonic_time()

    # Yields control to the next middleware (or Repo operation) in the chain
    # before resuming here.
    {result, _updated_resolution} = yield(resource, resolution)

    duration = System.monotonic_time() - start_time
    Logger.info("#{resolution.action} took #{duration}ns")

    result
  end
end
Important: When using process/2, you must call yield/2 to continue the middleware chain.
Halting Execution
Return {:halt, value} to stop the middleware chain and return a value immediately:
defmodule RequireAuth do
  use EctoMiddleware

  @impl EctoMiddleware
  def process(resource, resolution) do
    if authorized?(resolution) do
      {result, _} = yield(resource, resolution)
      result
    else
      {:halt, {:error, :unauthorized}}
    end
  end

  defp authorized?(resolution) do
    get_private(resolution, :current_user) != nil
  end
end
Guards for Operation Detection
EctoMiddleware.Utils provides guards to detect operations, especially useful for :insert_or_update:
defmodule ConditionalMiddleware do
  use EctoMiddleware

  @impl EctoMiddleware
  def process_before(changeset, resolution) when is_insert(changeset, resolution) do
    {:cont, add_created_metadata(changeset)}
  end

  def process_before(changeset, resolution) when is_update(changeset, resolution) do
    {:cont, add_updated_metadata(changeset)}
  end

  def process_before(changeset, _resolution) do
    {:cont, changeset}
  end
end
Return Value Conventions
For process_before/2 and process_after/2
Returning bare values from middleware is supported, but to be explicit, return one of:
	{:cont, value} - Continue to next middleware
	{:halt, value} - Stop execution, return value
	{:cont, value, updated_resolution} - Continue with updated EctoMiddleware.Resolution.t/0
	{:halt, value, updated_resolution} - Stop with updated EctoMiddleware.Resolution.t/0

Bare values are always treated as {:cont, value}.
Migration from V1
V1 middleware continue to work but emit deprecation warnings.
Key differences:
	Use EctoMiddleware.Repo instead of EctoMiddleware in Repo modules
	Implement process_before/2, process_after/2, or process/2 instead of the deprecated middleware/2
	No need for EctoMiddleware.Super marker
	Return {:cont, value} or {:halt, value} instead of bare values

See the Migration Guide for details.
Silencing Deprecation Warnings
During migration, you can silence warnings temporarily:
# In config/config.exs
config :ecto_middleware, :silence_deprecation_warnings, true
This should only be used during migration - deprecated APIs will be removed in v3.0.

Stubs out the necessary behaviours and imports for implementing an EctoMiddleware
middleware.
For backwards compatibility, if used in an Ecto.Repo module, it will emit a deprecation
warning and delegate to use EctoMiddleware.Repo instead. This behaviour will be removed in v3.0.

  


        

      


  

    
EctoMiddleware.Engine 
    



      
Internal engine for validating and executing middleware chains.
See EctoMiddleware for information on writing middleware. This module is used
internally by EctoMiddleware.Repo to execute middleware chains.
Silencing Deprecation Warnings
During migration from v1 to v2, you may want to silence deprecation warnings.
This can be done via application configuration:
# In config/config.exs
config :ecto_middleware, :silence_deprecation_warnings, true
This will suppress all deprecation warnings from EctoMiddleware. Note that this
should only be used temporarily during migration - the deprecated APIs will be
removed in v3.0.
Middleware Execution
Middleware are executed in a chain. You can think of this as a matroshka doll, where each
middleware wraps the next one in the chain.
Each middleware is expected to call yield/2 to continue execution to the next middleware
in the chain, but may also choose to halt execution by not calling yield/2 and returning
a value directly.
For example, given the following middleware chain:
[LoggerMiddleware, AuthMiddleware, VirtualFieldMiddleware]
Execution proceeds as follows:
	c:Ecto.Repo.insert/2 is called.
	EctoMiddleware.process/2 is invoked on LoggerMiddleware.	It logs "Before insert".
	It calls yield/2 to continue execution.


	EctoMiddleware.process/2 is invoked on AuthMiddleware.	It checks authorization.
	It calls yield/2 to continue execution.


	EctoMiddleware.process/2 is invoked on VirtualFieldMiddleware.	It immediately calls yield/2 to continue execution.


	There aren't any more middleware to execute, so the super function (the actual c:Ecto.Repo.insert/2 call) is invoked.	The database insert occurs, returning {:ok, user}.


	Control returns to VirtualFieldMiddleware.	It resolves some virtual fields on the user struct.
	It returns the user struct w/ virtual fields up the chain.


	Control returns to AuthMiddleware.	It doesn't do anything further, so it returns the user struct w/ virtual fields up the chain.


	Control returns to LoggerMiddleware.process/2.	It logs "After insert: {:ok, user w/ virtual fields}".
	It returns the user struct up the chain.


	The caller of c:Ecto.Repo.insert/2 sees the function result as {:ok, user_with_virtual_fields}.

For simplicity, middleware authors can either implement the EctoMiddleware.process_before/2 or
EctoMiddleware.process_after/2 callbacks to only operate in the "before" or "after" phases.
Alternatively, you can implement EctoMiddleware.process/2 for full control, but you must
call yield/2 at the appropriate time. Not calling yield/2 will halt execution at that
middleware.
Important: yield/2 returns {result, updated_resolution}. You must destructure this
tuple when calling yield.
See the EctoMiddleware docs for more information.
Backwards Compatibility
Prior versions of EctoMiddleware (v1.x) had a different middleware contract and execution engine.
In those versions of the library, all middleware were expected to implement middleware/2 in a
middleware module (note: this was not a behaviour callback). This function was expected to take some
"resource" and return that "resource" transformed by the middleware.
Middleware executed exactly once, either before or after the super function, depending on their
position relative to EctoMiddleware.Super in the middleware list.
An example of this follows:
defmodule Repo do
  use EctoMiddleware

  @impl EctoMiddleware
  def middleware(:insert, _resource) do
    [BeforeMiddleware, EctoMiddleware.Super, AfterMiddleware]
  end
end

defmodule BeforeMiddleware do
  def middleware(resource, _resolution) do
    transform_before(resource)
  end
end

defmodule AfterMiddleware do
  def middleware(resource, _resolution) do
    transform_after(resource)
  end
end
The current version of EctoMiddleware (v2.x) supports v1 middleware for backwards compatibility.
This is implemented by dynamically replacing any v1 middleware with a v2 middleware that wraps
the v1 middleware and calls its middleware/2 function either before or after the super function
is executed.
This functionality is temporary and will be removed in v3.0, at which point all middleware
must implement the v2 middleware contract.
Using v1 middleware in this way will emit a deprecation warning. We strongly recommend updating
any existing v1 middleware to implement the v2 middleware contract to avoid this warning and
ensure compatibility with future versions of EctoMiddleware.

          @spec validate_middleware!([term()]) :: [term()]


      


Validates each middleware implements the required callbacks.
This function is executed prior to execution of any given middleware chain to ensure
that the specified middleware pipeline is valid.
Note: This function also wraps any v1 middleware to be compatible with the v2 middleware
execution engine. This functionality is temporary and will be removed in v3 when v1 middleware
is no longer supported.

          @spec yield(resource :: term(), resolution :: EctoMiddleware.Resolution.t()) ::
  {term(), EctoMiddleware.Resolution.t()}


      


Yields execution to the next middleware in the chain.
Returns a tuple of {result, updated_resolution} where the resolution may have been
updated during middleware execution (e.g., for V1 compatibility fields like before_output).
	If the result of a middleware's process/2 function is {:halt, value}, execution is halted.
	If there are no more middleware to execute, the super function is invoked.
	Otherwise, execution continues to the next middleware in the chain.


  


        

      


  

    
EctoMiddleware.Repo behaviour
    



      
Enables middleware support for Ecto.Repo modules.
Add use EctoMiddleware.Repo to your Repo to enable middleware pipelines for
database operations.
Setup
defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app
  use EctoMiddleware.Repo

  @impl EctoMiddleware.Repo
  def middleware(action, resource) when is_insert(action, resource) do
    [NormalizeEmail, HashPassword, AuditLog]
  end

  def middleware(action, resource) when is_delete(action, resource) do
    [SoftDelete, AuditLog]
  end

  def middleware(_action, _resource) do
    [AuditLog]
  end
end
Defining Middleware
Implement the middleware/2 callback to specify which middleware run for each operation.
Pattern Matching on Actions
The first argument is the action atom (function name without arity):
@impl EctoMiddleware.Repo
def middleware(:insert, _), do: [ValidateEmail]

def middleware(:get, _), do: [EnrichData]

def middleware(:delete, _), do: [SoftDelete]
Available actions:
	Read: :get, :get!, :get_by, :get_by!, :one, :one!, :all, :reload, :reload!, :preload
	Write: :insert, :insert!, :update, :update!, :delete, :delete!, :insert_or_update, :insert_or_update!

Pattern Matching on Resources
The second argument is the resource being operated on:
# Specific schema
def middleware(:insert, %User{}), do: [NormalizeEmail, HashPassword]

# Multiple schemas with same middleware
def middleware(:insert, %{__struct__: schema})
  when schema in [User, Admin], do: [AuditLog]

# Changesets
def middleware(:update, %Ecto.Changeset{data: %User{}}), do: [CheckOwnership]
Using Guards
You can use EctoMiddleware.Utils guards in your middleware/2 definitions:
@impl EctoMiddleware.Repo
def middleware(action, resource) when is_insert(action, resource) do
  [SetCreatedAt, AuditLog]
end

def middleware(action, resource) when is_update(action, resource) do
  [SetUpdatedAt, AuditLog]
end

def middleware(_action, _resource), do: []
See EctoMiddleware.Utils for available guards.
Default Middleware
Always provide a catch-all clause:
def middleware(_action, _resource), do: []
Middleware Execution
Middleware execute in the order specified. If any middleware returns {:halt, value},
execution stops immediately.

          @type action() ::
  :all
  | :delete!
  | :delete

  | :get

  | :update


      



  



  
    
      
    
    
      middleware()



        
          
        

    

  


  

      

          @type middleware() :: module()


      



  



  
    
      
    
    
      resource()



        
          
        

    

  


  

      

          @type resource() ::
  %Ecto.Queryable{}
  | %Ecto.Changeset{}
  | %{__meta__: Ecto.Schema.Metadata}
  | {%Ecto.Queryable{}, Keyword.t()}

Enables the ability for a given Ecto.Repo to define and execute middleware.

          @spec middleware(repo :: module(), action(), resource()) :: [middleware()]


      


Returns the configured middleware for the given repo.

Resolution struct that tracks middleware execution context and state.
The resolution is passed to every middleware callback and contains information about
the current operation, remaining middleware, and provides storage for sharing data
between middleware.
Core Fields
	:repo - The Repo module executing the operation (e.g., MyApp.Repo)
	:action - The action being performed (e.g., :insert, :update, :get)
	:args - The full arguments list passed to the Repo function
	:middleware - The list of remaining middleware to execute
	:entity - The primary entity/resource being operated on (changeset, struct, query, etc.)
	:private - Map for passing data between middleware (use put_private/3 and get_private/2)

Using Private Storage
The :private field is a map for storing arbitrary data that middleware need to share:
defmodule SetContext do
  use EctoMiddleware

  def process_before(changeset, resolution) do
    resolution = put_private(resolution, :user_id, 123)
    resolution = put_private(resolution, :timestamp, DateTime.utc_now())
    {:cont, changeset, resolution}
  end
end

defmodule UseContext do
  use EctoMiddleware

  def process_after(result, resolution) do
    user_id = get_private(resolution, :user_id)
    timestamp = get_private(resolution, :timestamp)
    Logger.info("User #{user_id} at #{timestamp}")
    {:cont, result}
  end
end
Updating Resolution
To update the resolution and pass it forward, return a 3-tuple:
def process_before(changeset, resolution) do
  resolution = put_private(resolution, :key, :value)
  {:cont, changeset, resolution}
end
V1 Compatibility Fields (Deprecated)
The following fields exist for V1 backwards compatibility and will be removed in v3.0:
	:before_input - Original resource before middleware transformations
	:before_output - Resource after before middleware, before database operation
	:after_input - Raw database result before after middleware
	:after_output - Final result after all middleware
	:before_middleware - V1 before middleware list
	:after_middleware - V1 after middleware list

Do not use these fields in new code. Use :private storage instead.

          @type t() :: %EctoMiddleware.Resolution{
  action: term(),
  after_input: term(),
  after_middleware: term(),
  after_output: term(),
  args: term(),
  before_input: term(),
  before_middleware: term(),
  before_output: term(),
  entity: term(),
  middleware: term(),
  private: term(),
  repo: term()
}

          @spec get_private(t(), key :: atom(), default :: term()) :: term()


      


Retrieves a value from the resolution's private storage.
Returns the value if found, otherwise returns the provided default (or nil).
Examples
user_id = get_private(resolution, :user_id)
#=> 123

context = get_private(resolution, :context, %{})
#=> %{tenant: "acme"}

missing = get_private(resolution, :missing)
#=> nil

          @spec put_private(t(), key :: atom(), value :: term()) :: t()


      


Stores a key-value pair in the resolution's private storage.
This is useful for passing data between middleware in the chain.
Example
resolution = put_private(resolution, :user_id, 123)
resolution = put_private(resolution, :context, %{tenant: "acme"})

This is the deprecated super middleware for EctoMiddleware versions 1.x.
It exists solely for backward compatibility with middleware written for v1.x.
See documentation for EctoMiddleware.Engine for more details.

Utility functions and guards for working with middleware operations.
Guards for Operation Detection
Use guards in middleware to detect operation types, especially useful for
insert_or_update which can be either an insert or update:
defmodule SmartTimestamp do
  use EctoMiddleware

  def process_before(changeset, resolution) do
    changeset = cond do
      is_insert(changeset, resolution.action) ->
        changeset
        |> put_change(:created_at, DateTime.utc_now())
        |> put_change(:updated_at, DateTime.utc_now())

      is_update(changeset, resolution.action) ->
        put_change(changeset, :updated_at, DateTime.utc_now())

      true ->
        changeset
    end

    {:cont, changeset}
  end
end
Available Guards
	is_read/2 - Matches read operations (get, get!, all, etc.)
	is_write/2 - Matches write operations (insert, update, delete, etc.)
	is_insert/2 - Matches inserts (including insert_or_update with new records)
	is_update/2 - Matches updates (including insert_or_update with existing records)
	is_delete/2 - Matches delete operations
	is_preload/2 - Matches preload operations

How insert_or_update Detection Works
For insert_or_update actions, the guards inspect the changeset's __meta__.state:
# New record (insert)
changeset.data.__meta__.state == :built

# Existing record (update)
changeset.data.__meta__.state == :loaded
Result Transformation
Use apply/3 to transform Ecto results while preserving their structure:
defmodule EnrichUsers do
  use EctoMiddleware

  def process_after(result, resolution) do
    enriched = Utils.apply(result, resolution, fn user ->
      %{user | full_name: "#{user.first_name} #{user.last_name}"}
    end)

    {:cont, enriched}
  end
end
apply/3 handles all common Ecto return types:
	{:ok, value} → {:ok, transformed}
	{:error, reason} → unchanged
	[values] → [transformed values]
	nil → unchanged
	{count, list} → {count, [transformed values]}
	Bare values → transformed

See apply/3 documentation for complete examples.

          @spec apply(result :: term(), context :: struct(), fun :: (term() -> term())) ::
  term()


      


Applies a function to the "inner" value of an Ecto result, handling all common return types.
The second argument is reserved for future use and currently ignored. You can pass the
resolution struct or any placeholder value.
Return Type Handling
	Input	Output
	{:ok, value}	{:ok, fun.(value)}
	{:error, reason}	{:error, reason} (unchanged)
	[values]	Enum.map(values, fun)
	nil	nil (unchanged)
	{count, list}	{count, Enum.map(list, fun)}
	{count, nil}	{count, nil} (unchanged)
	bare value	fun.(value)

Examples
# In process_after/2
def process_after(result, resolution) do
  {:cont, Utils.apply(result, resolution, &Map.put(&1, :enriched, true))}
end

# {:ok, struct} -> {:ok, enriched_struct}
Utils.apply({:ok, user}, resolution, &Map.put(&1, :enriched, true))
#=> {:ok, %User{enriched: true, ...}}

# {:error, changeset} -> {:error, changeset} (unchanged)
Utils.apply({:error, changeset}, resolution, &Map.put(&1, :enriched, true))
#=> {:error, changeset}

# [structs] -> [enriched_structs]
Utils.apply([user1, user2], resolution, &Map.put(&1, :enriched, true))
#=> [%User{enriched: true}, %User{enriched: true}]

# nil -> nil
Utils.apply(nil, resolution, &Map.put(&1, :enriched, true))
#=> nil

# {count, list} from update_all/delete_all
Utils.apply({3, [user1, user2, user3]}, resolution, &Map.put(&1, :processed, true))
#=> {3, [%User{processed: true}, ...]}

Guard that matches delete actions.
Matches the action atoms for c:Ecto.Repo.delete/2 and c:Ecto.Repo.delete!/2.

Guard that matches insert actions.
Matches the action atoms for c:Ecto.Repo.insert/2 and c:Ecto.Repo.insert!/2.
Also matches c:Ecto.Repo.insert_or_update/2 and c:Ecto.Repo.insert_or_update!/2 when
the record is new (changeset's __meta__.state is :built).

Guard that matches the preload action.
Matches the action atom for c:Ecto.Repo.preload/3.

Guard that matches read actions.
Matches the action atoms for the following Ecto.Repo callbacks:
c:Ecto.Repo.get/3, c:Ecto.Repo.get!/3, c:Ecto.Repo.get_by/3, c:Ecto.Repo.get_by!/3,
c:Ecto.Repo.one/2, c:Ecto.Repo.one!/2, c:Ecto.Repo.all/2, c:Ecto.Repo.reload/2,
c:Ecto.Repo.reload!/2, c:Ecto.Repo.preload/3

Guard that matches update actions.
Matches the action atoms for c:Ecto.Repo.update/2 and c:Ecto.Repo.update!/2.
Also matches c:Ecto.Repo.insert_or_update/2 and c:Ecto.Repo.insert_or_update!/2 when
the record already exists (changeset's __meta__.state is :loaded).

Guard that matches write actions (anything not a read).
Matches the action atoms for the following Ecto.Repo callbacks:
c:Ecto.Repo.insert/2, c:Ecto.Repo.insert!/2, c:Ecto.Repo.update/2, c:Ecto.Repo.update!/2,
c:Ecto.Repo.delete/2, c:Ecto.Repo.delete!/2, c:Ecto.Repo.insert_or_update/2,
c:Ecto.Repo.insert_or_update!/2

